
/* ##################################################################
/* ############         APDISPLAY.AML             ###################
/* ##################################################################

/* ==================================================================
/* PURPOSE:
/* Displays different thematic map series in AP.  Legends are
/* displayed, with the key files being generated automatically if
/* they are not created yet.
/* Titles for each thematic series can be shown.  Time is also
/* displayed.  This time is computed from the elapsed time between
/* CASC2D maps, and the map number in the series.
/*
/* ==================================================================
/* ARGUMENTS:
/*  None
/*
/* ==================================================================
/* VARIABLES:
/*  Passed from the main program (DS.aml) as global
/*  variables.
/*
/* ==================================================================
/* CALLS:
/*  TakeScreenShot.aml
/*
/* ==================================================================
/* ROUTINES:
/*   Legend    -- Creates key files and displays map legends
/*      according to the given remap table and shadeset
/*   TitleText -- Writes map titles (specified in the Parameter file)
/*   TimeText  -- Writes the event simulation time
/*   SetLayoutVariables -- Specifies position of legend, title, sub-
/*      title, simulation time and maps in a given layout
/*
/* ==================================================================
/* HISTORY:
/*   August 18, 2001:  This part of the code is taken from the main
/*  program (DisplaySeries.aml) and becomes a stand alone aml.
/*  November 16, 2001: Display2.aml, Display3.aml....Display6.aml
/*  are re-writen more generically in only one AML (Display.aml).
/*
/* ==================================================================

/* Displays a red line around the AP window
LINECOLOR 2
LINESIZE 0.05

/* I want a white background.  Instead of changing the canvas color
/* I will just put a white patch as background
SHADETYPE COLOR
SHADECOLOR 1
&SV CanvasColor = [SHOW CANVASCOLOR]
&TYPE Canvas Color is %CanvasColor%

/* SPECIFY ARCPLOT DISPLAY CONFIGURATION
&IF %.MapsPerFrame% EQ 1 OR %.MapsPerFrame% EQ 4 &THEN

&DO
DISPLAY 9999 SIZE CANVAS 720 600 /* Landscape

PAGESIZE 11 8.5 /* Landscape
/* &IF %CanvasColor% EQ BLACK &THEN PATCH 0 0 11 8.5
BOX 0 0 11 8.5 /* Landscape

&END
&ELSE

&DO
DISPLAY 9999 SIZE CANVAS 530 720 /* Portrait
PAGESIZE 8.5 11 /* Portrait
/*&IF %CanvasColor% EQ BLACK &THEN PATCH 0 0 8.5 11
BOX 0 0 8.5 11 /* Portrait

&END

/* LEGEND AND TITLES WILL NOT CHANGE ALONG THE SIMULATION SO THEY ARE
/* DISPLAYED ONLY ONCE AT THE BEGINING (they won't flicker every time
/* the display is refreshed and the display will be faster)

/* Set variables depending on the number of maps per frame
&CALL SetLayoutVariables

/* Creates Legends and Titles in layout
&CALL Legend
&CALL TitleText

/* DISPLAY THEMATIC MAP SERIES

&DO FrameNumber = %.StartFrame% &TO %.EndFrame% &BY %.FrameStep%

GNDS TRANSPARENT /* NODATA cells will be displayed transparently.
/* that I can see the map legend

&SV timesec [calc %FrameNumber% * %.ElapsedTime%]
&SV timemin [calc %timesec% / 60]
&CALL TimeText

&TYPE Displaying frame Number: %FrameNumber%

&DO n = 1 &TO %.MapsPerFrame%
MAPLIMITS [UNQUOTE[VALUE MLimits%n%]]
MAPEXTENT %.wPath%[VALUE .Name%n%]%FrameNumber%
MAPPOSITION LR LR
SHADEDELETE ALL
SHADESET [VALUE .ShdSet%n%]
GRIDSHADES ~

%.wPath%[VALUE .Name%n%]%FrameNumber% VALUE [VALUE .RMT%n%]
LINECOLOR 2
LINESIZE 0.04
/*ARCS h:\gismaps\3hydromodel\30\drainagedata\basins_polys

&END

&PAUSE &SECONDS %.SecsPause%

&IF %.CreateMovie% = 1 &THEN &DO
&RUN TakeScreenShot
&TYPE Done: Screen shot of frame Number: %FrameNumber%

&END

&END

&RETURN

/* ##################################################################
/* ############       ROUTINE LEGEND              ###################
/* ##################################################################

/* CREATES FILES TO STORE THE KEY FILES.
/* KEY FILES ARE DERIVED FROM REMAP TABLES

&ROUTINE Legend

&DO k = 1 &TO %.MapsPerFrame%

&SV fileunit%k% [OPEN %.wPath%[value .Name%k%].key openstat -write]

&IF %openstat% <> 0 &THEN
&DO

&SV str Could not create key_file ...; &CALL bailout
&END

/* DATA WILL BE TAKEN FROM THE REMAP TABLE (DEFINED BY THE USER)
/* AND WRITEN TO THE TEMPORAL KEY FILE WITH THE CORRESPONDING
/* KEY FILE FORMAT (see "Drawing Key Legend" help)

&SV eof 0
&SV RemapFile%k% [open [value .RMT%k%] openstat -read]
&SV line [UNQUOTE [READ [value RemapFile%k%] eof]]

/* I ASSUME THAT THE USER DOESN'T HAVE MORE THAN 100 CLASSES!!!
/* I WILL ONLY WRITE THE UPPER LIMIT AS THE TEXT CLOSE TO THE KEY
/* IF YOU WANT TO WRITE THE WHOLE INTERVAL --> CHANGE NEXT LINES
/* OF CODE TO READ AND WRITE THE 1st AND 2nd ARGUMENT OF %line%

&DO i = 1 &TO 100 &WHILE %eof% NE 102
/* 102 is returned value for EOF
&SV NULL [WRITE [VALUE fileunit%k%] .%i% ]

&SV key%i% := [QUOTE < [EXTRACT 2 %line%]]
&SV NULL [WRITE [VALUE fileunit%k%] [value key%i%]]
&SV line [UNQUOTE [READ [value RemapFile%k%] eof]]

&END

&SV NULL [CLOSE [VALUE fileunit%k%]]

&SV closestat = [CLOSE [VALUE RemapFile%k%]]
&END

/* TODO:CHANGE THE NEXT LINES WITH THE FONTS AND SIZES OF YOUR LIKING
/* NEXT SPECIFICATIONS WILL BE COMMON FOR ALL LEGENDS

/* Standandard hardware device color index:
/* 1:white; 2:red; 3:green; 4:blue; 5:cyan; 6:magenta; 7:yellow

SHADETYPE COLOR
TEXTCOLOR 1 /* Legend text color: black
TEXTFONT 'Times Bold' /* Legend Font:smooth panel
TEXTSIZE 0.23 0.23 /* Legend text size: .2 inches
TEXTSPACING 0
TEXTQUALITY PROPORTIONAL
LINESIZE 0.01 /* Line width around key box

LINECOLOR 0 /* Line color (black) around key box
KEYBOX .25 .18 /* Key box width and height (inches)
KEYSEPARATION 0.05 0.03 /* Key dist. to text and between keys

&DO j = 1 &TO %.MapsPerFrame%
/*BOX [UNQUOTE[VALUE Area%j%]]
SHADEDELETE ALL
KEYAREA [UNQUOTE [value Area%j%]]
SHADESET [value .ShdSet%j%]
KEYSHADE %.wPath%[value .Name%j%].key
&SV NULL [DELETE %.wPath%[value .Name%j%].key] /* Delete key file

&END

&RETURN

/* ##################################################################
/* ############       ROUTINE TITLE_TEXT          ###################
/* ##################################################################

/* CREATES THE TITLES AND SUBTITLES CORRESPONDING TO EACH OF THE
/* THEMATIC SERIES -- TITLES ARE USER DEFINED IN THE PARAMETER FILE

&ROUTINE TitleText

TEXTCOLOR 2 /* Title in red
TEXTSIZE 0.4 0.4 /* Title font height & width: 0.4 inches
TEXTFONT 'Courier bold' /*
TEXTSPACING 0
TEXTQUALITY PROPORTIONAL

/* MOVE: specifies a coordinate point used to position the cursor and
/* to position text drawn with TEXT and TEXTFILE. Specify only the
/* lower left corner of the title and subtitle block.

&DO m = 1 &TO %.MapsPerFrame%
MOVE [VALUE xTitlePos%m%] [VALUE yTitlePos%m% ]
TEXT [value .MapTitle%m%]
MOVE [VALUE xTitlePos%m%] [CALC [VALUE yTitlePos%m%] - 0.4]
TEXT [value .MapSubTitle%m%]

&END

&RETURN

/* ##################################################################
/* ############       ROUTINE TIME_TEXT          ####################
/* ##################################################################

/* CALCULATES SIMULATION TIME AND DISPLAYS IT ON THE LAYOUT

&ROUTINE TimeText

/* Needed to put a black box on top of the time so that the last
/* time is not shown (the display window is not being refreshed )

SHADETYPE COLOR
SHADECOLOR 0

/* (xmin, ymin) has been specified in the SetVariables routine
&SV xmax = [CALC %xmin% + 2]

&SV ymax = [CALC %ymin% + 0.55]

PATCH %xmin% %ymin% %xmax% %ymax%

MOVE [CALC %xmin% + 0.25] [CALC %ymin% + 0.25]

TEXTSIZE 0.40 0.40
TEXTSPACING 0
TEXTQUALITY PROPORTIONAL
TEXTFONT 'omega bold'
TEXT 't = '
TEXT [QUOTE [format '%1%' %timemin%]]
MOVE [CALC %xmin% + 1.8] [CALC %ymin% + 0.25]
TEXTFONT 'times' /* ??????????????????????
TEXT ' min.' /* How to write the whole thing

/* with only one TEXT command ?
&RETURN

/* ##################################################################
/* ############       ROUTINE SET_VARIABLES      ####################
/* ##################################################################

&ROUTINE SetLayoutVariables

&IF %.MapsPerFrame% EQ 1 &THEN
&DO

&SV MLimits1 = ' 0 0 11 8.5 '
&SV Area1 = '0.2 5.0 4.3 8.4'
&SV xTitlePos1 = 3.5
&SV yTitlePos1 = 7.35
&SV xmin = 6.75
&SV ymin = 0.75

&END

&IF %.MapsPerFrame% EQ 2 &THEN /*   ____________
&DO /*  |           |

&SV MLimits1 = '0.1 5.4 8.4 10.9' /*  |     1     |
&SV MLimits2 = '0.1 0.1 8.4 5.5' /*  |___________|
&SV Area1 = '0.1 8.9 2.85 10.65' /*  |           |
&SV Area2 = '0.1 3.6 2.85 5.35' /*  |     2     |
&SV xTitlePos1 = 2.4 /*  |___________|
&SV yTitlePos1 = 10.6
&SV xTitlePos2 = %xTitlePos1%
&SV yTitlePos2 = [CALC %yTitlePos1% - 5.3 ]
&SV xmin = 4.20
&SV ymin = 0.1

&END

/* this one needs more work. Check values
&IF %.MapsPerFrame% EQ 3 &THEN

&DO
&SV MLimits1 = '0.0 7.4 8.5 11.0'
&SV MLimits2 = '0.0 3.7 8.5 7.3'
&SV MLimits3 = '0.0 0.0 8.5 3.6'
&SV Area1 = '0.1 8.9 2.85 10.65'
&SV Area2 = '0.1 3.6 2.85 5.35'
&SV Area3 = '0.1 3.6 2.85 5.35'
&SV xTitlePos1 = 5.75 /*   ____________
&SV yTitlePos1 = 0.9 /*  |           |

&SV xTitlePos2 = 5.75 /*  |     1     |
&SV yTitlePos2 = 6.3 /*  |___________|
&SV xTitlePos3 = 6.3 /*  |           |
&SV yTitlePos3 = 5 /*  |     2     |
&SV xmin = 2.20 /*  |___________|
&SV ymin = 0.1 /*  |           |

&END /*  |     3     |
/*  |___________|

&IF %.MapsPerFrame% EQ 4 &THEN
&DO

&SV MLimits1 = '0 4.25 5.5 8.5'
&SV MLimits2 = '5.5 4.25 11 8.5'
&SV MLimits3 = '0 0 5.5 4.25'
&SV MLimits4 = '5.5 0 11 4.25'
&SV Area1 = '0.05 6.95 2.05 8.45'
&SV Area2 = '5.40 6.95 7.40 8.45'
&SV Area3 = '0.05 2.75 2.05 4.25'
&SV Area4 = '5.40 2.75 7.40 4.25'
&SV xTitlePos1 = 2.70
&SV yTitlePos1 = 8.05 /*   _______________________|
&SV xTitlePos2 = 8.10 /*  |           |           |
&SV yTitlePos2 = 8.05 /*  |     1     |     2     |
&SV xTitlePos3 = 2.70 /*  |___________|___________|
&SV yTitlePos3 = 3.85 /*  |           |           |
&SV xTitlePos4 = 8.10 /*  |     3     |     4     |
&SV yTitlePos4 = 3.85 /*  |___________|___________|
&SV xmin = 3.0
&SV ymin = 0.05

&END

&IF %.MapsPerFrame% EQ 6 &THEN
&DO

&SV MLimits1 = '0.0 7.2 4.2 10.8' /*Landscape: 0 4.2 3.6 8.4
&SV MLimits2 = '4.2 7.2 8.4 10.8' /*Landscape: 3.6 4.2 7.2 8.4
&SV MLimits3 = '0.0 3.6 4.2 7.2' /*Landscape: 7.2 4.2 10.8 8.4
&SV MLimits4 = '4.2 3.6 8.4 7.2' /*Landscape: 0 0 3.6 4.2
&SV MLimits5 = '0.0 0.0 4.2 3.6' /*Landscape: 3.6 0 7.2 4.2
&SV MLimits6 = '4.2 0.0 8.4 3.6' /*Landscape: 7.2 0 10.8 4.2
&SV Area1 = '0.05 9.15 1.65 10.85'
&SV Area2 = '4.2 9.15 5.8 10.85'
&SV Area3 = '0.05 5.55 1.65 7.25'
&SV Area4 = '4.2 5.55 5.8 7.25'
&SV Area5 = '0.05 1.95 1.65 3.65'
&SV Area6 = '4.2 1.95 5.8 3.65'
&SV xTitlePos1 = 2 /*   _______________________
&SV yTitlePos1 = 10.2 /*  |           |           |
&SV xTitlePos2 = 6.15 /*  |     1     |     2     |
&SV yTitlePos2 = 10.2 /*  |___________|___________|
&SV xTitlePos3 = 2 /*  |           |           |
&SV yTitlePos3 = 6.6 /*  |     3     |     4     |
&SV xTitlePos4 = 6.15 /*  |___________|___________|
&SV yTitlePos4 = 6.6 /*  |           |           |
&SV xTitlePos5 = 2 /*  |     5     |     6     |
&SV yTitlePos5 = 3.0 /*  |___________|___________|
&SV xTitlePos6 = 6.15
&SV yTitlePos6 = 3.0
&SV xmin = 2.20
&SV ymin = 0.05

&END

&RETURN


